


ELBUG

SUPPLEMENT TO BEEBUG

JUNE 1985 VOLUME 2 ISSUE 7



FILE HANDLING (Part 3)

By Geoff Bains

Geoff Bains continues his look at mastering the file handling capabilities of the Electron and creates a simple card index type of database.

In the last two parts of 'File Handling' we have seen how to output data to cassette (or disc) with the OPENOUT and PRINT# commands and input it again with OPENIN and INPUT#. We have also seen how a very simple index can be kept on cassette and useful data squeezed from it. This month we will take a look at how you can edit files that you have built up in this way.

The best way to illustrate the methods involved is by example. At the end of the article you will find a complete program listing of a useful card index type database program. This will allow you to specify the format of the records you want to store - the number of fields and the

title for each field - and then enter data for as many records as you want. Then a file can be browsed through, edited, more records added, and records deleted. Finally you can search through the whole file for the occurrence of any string of characters.

The only limits to your file are the lengths of various parts. The field titles must not be longer than ten letters and there can be no more than ten fields to a record. The fields themselves must be no longer than 25 letters. These limitations are inbuilt in the program itself. They are largely due to the constraints of displaying a record on a mode 6 screen. No checks are made in the program to reject →

ELBUG CASSETTE ELBUG CASSETTE ELBUG CASSETTE ELBUG CASSETTE ELBUG CASSETTE ELBUG CASS

The June ELBUG magazine cassette is packed with goodies for the Electron. As well as the programs from this month's BEEBUG - the Stonehenge display, multiple windows utility, the binary and hash searches from the Workshop, the logic examples from the Beginners article, and the excellent Trackman game - there is also the Cardbox database program from File Handling in this ELBUG supplement, along with a data file to get you started, and a bonus two player game called Swan Trap.

inputs that exceed these limits. This is something you could add for yourself.

The program works by keeping the entire file in memory. When a file is loaded for inspection, editing, and so on, all the records are loaded at once, into arrays in RAM. The number of records in a file is therefore limited. This is due to the limited memory of an Electron. Mode 6 has been used for all the program functions to release the maximum memory but the number of records is still limited to about 100. The exact number you will be able to fit in will depend on the records themselves. A bit of experimentation is called for here.

The cassette (or disc) file has two 'extra' items at its start. These are the number of records in the file and the number of fields in each record. The next group of items in the cassette file is the titles of the fields. The rest of the cassette file consists of a list of the field contents of each record in turn. In the main part the data is actually processed only in arrays, in memory, and not in the storage file itself. However, there are several 'tricks' used that make a program such as this easier and more friendly to use.

When a file is being inspected or edited, a check is kept to see if any changes are made to the file (using the flag, change%). Only if changes are made, is the altered file saved to cassette at the end of a session. Records can be added either at the end of a file or at a particular position in the middle. When a record is being edited the old contents of each field are presented on the screen and these can be accepted by the edited version simply by pressing Return. Otherwise a new entry is typed in.

To help you find your way around the program, here is a list of all the procedures and functions along with their purpose. The program is quite well structured so this list really sums up the whole operation.

PROColdfile Inputs filename, reads in the number of records and number of fields. Dimensions arrays accordingly.

PROCfilesize Inputs filename, and details of file to be created. Dimensions arrays.

PROCnewfile Calls PROCentry for each record.

PROCentry Inputs contents of each field of a specified record against title.

PROCshow Displays a specified record.

PROCrecord Alters record pointer to jump to a record.

PROCsave Saves the complete file with a new, or the existing, filename.

PROCdelete Deletes a record and moves all above it down. Decrements the number of records.

PROCadd Moves records up and adds a new record using PROCentry. Increments number of records.

PROCsearch Inputs search string. Repeatedly searches each record using FNsearchrec, displaying each record where a match is found, using PROCshow.

FNsearchrec Searches through file until a match is found for the search string.

FNmenu Displays the main menu and inputs the choice.

Next month we will look at the more important procedures in this program in more detail. Meanwhile you have a useful card index program to store away all your dark secrets!

```
10 REM Program CARDBOX
20 REM Version E0.1
30 REM Author Geoff Bains
40 REM ELBUG June 1985
50 REM Program subject to copyright
55 :
56 ON ERROR GOTO 2690
60 :
```



```

100 MODE6
110 VDU19,0,4;0;
120 MAX=100
130 change%=FALSE
140 records%=MAX
150 F$=""
160 REPEAT
170 IF change% THEN PROCsave
180 A=FNmenu
190 IF A=1 THEN PROCfilesize:PROCnewfile
200 IF (A=2 OR A=3) AND F$="" THEN PROColdfile
210 IF A=2 AND F$<>"" THEN PROCshow(1,TRUE)
220 IF A=3 PROCsearch
230 UNTIL A=9
240 MODE 6
250 END
260 :
1000 DEF PROColdfile
1010 CLS
1020 INPUT TAB(10,10)"Filename ",F$
1030 chan=OPENIN F$
1040 INPUT #chan,records%,fields%
1050 DIM A$(MAX,fields%)
1060 FOR r%=0 TO records%
1070 FOR f%=0 TO fields%
1080 INPUT #chan,A$(r%,f%)
1090 NEXT f%,r%
1100 CLOSE #chan
1110 ENDPROC
1120 :
1130 DEF PROCfilesize
1140 CLS
1150 INPUT TAB(10,1)"Filename ",F$
1160 INPUT TAB(10,3)"Number of fields ",F:fields%=F-1
1170 IF fields%>9 VDU7:PRINT TAB(10,3)"TOO MANY FIELDS";SPC(8):T%=TIME:REPEAT UNTIL TIME>T%+100:GOTO 1160
1180 DIM A$(MAX,fields%)
1190 FOR f%=0 TO fields%
1200 PRINT
1210 PRINT " Title of field ";f%+1;
1220 INPUT A$(0,f%)
1230 NEXT f%
1240 CLS
1250 ENDPROC
1260 :
1270 DEF PROCnewfile
1280 flag%=FALSE
1290 FOR I%=1 TO records%
1300 PROCentry(I%,TRUE,FALSE)
1310 SOUND 1,-15,150,5
1320 IF flag% THEN temp%=records%:records%=I%-1:I%=temp%
1330 NEXT I%
1340 PROCsave
1350 ENDPROC

```

```

1360 :
1370 DEF PROCentry(NUM,TEST%,old%)
1380 CLS
1390 PRINT TAB(10,1)F$,"Record ";NUM
1400 IF NOT old% PRINT TAB(3,23)"Input @ for first field to finish"
1410 FOR f%=0 TO fields%
1420 PRINT TAB(5,f%+5)A$(0,f%):IF old% PRINT TAB(15,f%+5)A$(NUM,f%)
1430 INPUT TAB(15,f%+5)IN$
1440 IF TEST% AND f%=0 AND IN$="@" THEN flag%=TRUE:f%=fields%
1450 IF NOT old% OR IN$<>"" A$(NUM,f%)=IN$
1460 NEXT f%
1470 ENDPROC
1480 :
1490 DEF PROCshow(r%,options%)
1500 REPEAT
1510 CLS
1520 PRINT TAB(10,1)F$,"Record ";r%
1530 FOR f%=0 TO fields%
1540 PRINT TAB(5,f%+5)A$(0,f%);TAB(15,f%+5)A$(r%,f%)
1550 NEXT f%
1560 IF NOT options% THEN G$="M":GOTO 1690
1570 IF r%=records% THEN VDU26:PRINT TAB(10,18)"Last record in ";F$:VDU7
1580 PRINTTAB(0,21)"N next record P preceeding record"
1590 PRINTTAB(0,22)"D delete record A add record"
1600 PRINTTAB(0,23)"C change record G goto record no."
1610 PRINTTAB(10,24)"M main menu";
1620 G$=CHR$(GET AND 223)
1630 IF G$="D" change%=TRUE:PROCdelete
1640 IF G$="A" change%=TRUE:PROCadd
1650 IF G$="C" change%=TRUE:PROCentry(r%,FALSE,TRUE)
1660 IF G$="N" r%=r%+1+(r%=records%)
1670 IF G$="P" r%=r%-1-(r%=1)
1680 IF G$="G" PROCrecord
1690 UNTIL G$="M" OR NOT options%
1700 ENDPROC
1710 :
1720 DEF PROCrecord
1730 SOUND1,-15,150,5
1740 PRINT TAB(10,18)SPC(25)
1750 INPUT TAB(10,18)"Record number ",r%
1760 IF r%>records% THEN r%=records%
1770 ENDPROC
1780 :
1790 DEFPROCsave
1800 CLS
1810 PRINT TAB(15,0)"Saving file"
1820 PRINT TAB(8,2)"(existing filename

```

```

: "F$;"
1830 PRINT TAB(2,8)"1      Save file wit
h existing filename"
1840 PRINT TAB(2,10)"2      Save file wi
th new filename"
1850 PRINT TAB(2,12)"3      Abort"
1860 REPEAT
1870 A=GET-48
1880 UNTIL A<4 AND A>0
1890 IF A=2 INPUT TAB(10,16)"New filena
me ",F$
1900 IF A=3 THEN ENDPROC
1910 chan=OPENOUT(F$)
1920 PRINT #chan,records%,fields%
1930 FOR r%=0 TO records%
1940 FOR f%=0 TO fields%
1950 PRINT #chan,A$(r%,f%)
1960 NEXT f%,r%
1970 CLOSE #chan
1980 VDU7:PRINT TAB(10,18)F$;" saved":c
hange%=FALSE:T%=TIME:REPEAT UNTIL TIME=T
%+50
1990 ENDPROC
2000 :
2010 DEF PROCdelete
2020 PRINT TAB(6,18)"Delete this record
(Y/N)"
2030 SOUND1,-15,150,5
2040 G$=CHR$(GET AND 223):IF G$<>"Y" TH
EN 2110
2050 FOR I%=r% TO records%-1
2060 FOR f%=0 TO fields%
2070 A$(I%,f%)=A$(I%+1,f%)
2080 NEXT f%,I%
2090 IF r%=records% THEN r%=r%-1
2100 records%=records%-1
2110 ENDPROC
2120 :
2130 DEFPROCadd
2140 IF records%>MAX THEN VDU7:ENDPROC
2150 PRINT TAB(8,18)"Insert or append
(I/A)"
2160 SOUND1,-15,150,5
2170 G$=CHR$(GET AND 223):IF G$="A" r%=
records%+1:GOTO2220
2180 FOR I%=records% TO r% STEP-1
2190 FOR f%=0 TO fields%
2200 A$(I%+1,f%)=A$(I%,f%)
2210 NEXT f%,I%
2220 records%=records%+1
2230 PROCentry(r%,FALSE,FALSE)
2240 ENDPROC
2250 :
2260 DEF PROCsearch
2270 CLS
2280 INPUT TAB(5,10)"Search string "S$
2290 IF S$="" VDU7:GOTO 2270

```

```

2300 I%=0
2310 I%=FNsearchrec(I%)
2320 IF (NOT found%) AND (I%=records%)
PRINT TAB(5,18)"No matches for ";S$:GOTO
2410
2330 REPEAT
2340 J%=I%
2350 I%=FNsearchrec(I%)
2360 PROCshow(J%,FALSE)
2370 IF found% PRINT TAB(5,18)"For next
match of ";S$ TAB(5,20)"Press space bar
":REPEAT UNTIL GET=32
2380 UNTIL I%=records%
2390 IF found% PROCshow(I%,FALSE)
2400 PRINT TAB(5,18)"No more matches fo
r ";S$
2410 VDU7:PRINT TAB(5,20)"Press space b
ar":REPEAT UNTIL GET=32
2420 ENDPROC
2430 :
2440 DEF FNsearchrec(r%)
2450 REPEAT
2460 r%=r%+1
2470 found%=FALSE
2480 FOR f%=0 TO fields%
2490 IF INSTR(A$(r%,f%),S$) THEN found%
=TRUE
2500 NEXT f%
2510 UNTIL found% OR r%=records%
2520 =r%
2530 :
2540 DEF FNmenu
2550 *FX15,0
2560 VDU 26,12
2570 PRINT TAB(16,2)"CARDBOX"
2580 PRINT TAB(16,3)"-----"
2590 PRINT TAB(10,10)"1      Create new
file"
2600 PRINT TAB(10,12)"2      View file"
2610 PRINT TAB(10,14)"3      Search file
"
2620 PRINT TAB(10,18)"9      Exit"
2630 PRINT TAB(7,23)"Press key of your
choice"
2640 REPEAT
2650 A=GET-48
2660 UNTIL A=1 OR A=2 OR A=3 OR A=9
2670 =A
2680 :
2690 ON ERROR OFF
2700 MODE 6
2710 IF ERR<>17 REPORT:PRINT " at line
";ERL
2720 END

```


Your Guide To BEEBUG

Just about all the BEEBUG programs and articles are, this month, useful to ELBUG members. There are of course some changes to be made to a few programs and some general comments that can be made on Beeb program conversion.

Conversion of Beeb programs is usually required for one of two reasons - sound and mode 7. As the conversion is usually for the same reasons, it is usually very straightforward.

The Electron has only one sound channel, instead of the BBC micro's three, and no control over the volume of the sound. The Electron is designed so that BBC micro programs making full use of the SOUND command will run on the Electron, though not, of course, producing the full effect. In most cases, then, the Beeb program can be left just as it is with no ill effects.

The Beeb's mode 7 display is of the same format as mode 6 but with a full range of colours and crude graphics. Mode 7 is not available on the Electron, but when the Electron encounters a MODE 7 command it defaults to a mode 6 screen. So, although you should, strictly speaking, change each occurrence of MODE 7 in a BBC program to MODE 6, this isn't really necessary because the Electron treats such commands as MODE 6 anyway. BBC micro programs that include a mode 7 display will therefore run on the Electron but often with a corrupted (and monochrome) display.

The one difference between the Electron and the BBC micro that you cannot alter the programs to account for is the difference in speed. The Electron runs considerably slower than the Beeb in many circumstances. This will affect some programs but not all. Where possible such changes as are possible to reduce the effects of this are given for ELBUG readers.

NEW BBC FROM ACORN

This piece of good news affects Elbug members as much as those in BEEBUG. It is always good to know that Acorn are apparently thriving, especially after recent troubles, and if you are thinking

of upgrading your Elk to something more powerful, what better than the model B+ Beeb.

STONEHENGE

This excellent visual program gives you and your Electron all the joys of Salisbury Plain with none of the train fares there.

The program will run exactly as it stands on the Electron with no changes needed.

MULTIPLE WINDOWS

Windows really come into their own when you can switch back and forth between different ones on the screen at once. That's a bit of a handful with BBC Basic but this program makes it all much easier.

There's nothing to be changed here for the Electron owner either.

COMAL

Comal is not yet available for the Electron but Acornsoft promises that it will be soon. The actual software is to be identical to the Beeb ROM we review here, but there have been holdups in manufacturing the Plus 1 cartridge in which it is to be marketed. Electron Comal should be in the shops by September. If you own one of the Electron ROM boards available then you can use the BBC version immediately.

MAKING THE MOST OF WORDWISE PLUS

Unfortunately this near revolutionary word processor is not available on the Electron.

BEGINNERS START HERE - All about logic

This month's beginner's spot looks at all the ins and outs of logic - TRUE and FALSE, IF-THENS, NOT, and so on. This can be one of the most confusing areas of programming but all is made crystal clear by Peter Lewis.

Everything that is said in this article and all the short programs in it are fully compatible with the Electron.

POSTBAG

More comments from the public at large. How about a few from ELBUG members?

GAMES REVIEWS

More's the shame, the two best games of this little selection, Caveman Capers and Manic Minor, do not run on the Electron and are not available in forms that do. However, the excellent Tempest is available in an Electron version and Starmaze will operate on an Electron (although one control - the TAB key on the Beeb - presents problems on an Elk!)

PENMAN PLOTTER

The Penman Plotter cannot work with the Electron because it relies on an RS232 interface for communication with the host micro. Until such time as an RS232 interface is produced for the Electron this review will remain only an interesting insight into a novel product.

WORKSHOP - Searching and Sorting (Part 3)

The Workshop continues this month with methods of searching data lists. Anyone interested in using their Electron to process any kind of numeric or string data will find this article invaluable.

Both the Binary Search program and that for the Hash Search work perfectly on the Electron.

ADVENTURE GAMES

The Saga of Erik the Viking will not run on an Electron because the lack of a mode 7 display prohibits the large amount of data required by this game.

EXTENDED DISC CATALOGUE

Even if you are fortunate enough to own an Acorn Plus 3, the extended disc catalogue will not work with your machine because of the different way in which the ADFS in the Plus 3 stores files in a different manner to the BBC micro DFS.

BEEBUG MUSIC COMPETITION

Although ELBUG members are, on the whole, equally as musical as BBC owners, the prizes for this competition are not compatible with the Electron.

MAKING MUSIC ON THE BEEB (Part 5)


In the last of his series on music Ian Waugh looks at sound effects available on the Beeb. Some of the ideas are equally applicable to the Electron. Trills and vibrato are both possible on the Electron but tremolo requires an amplitude envelope and ring modulation more than one channel, so these will not be operable.

HOME EDUCATION FROM ACORNSOFT

Spooky Manor is unfortunately not available on the Electron. However, Workshop is available to Electron owners for £9.20 so if you fancy a bit of homework check out the review in BEEBUG.

GRAPHITO AND TESSELATOR

One of the greatest assets of the Electron is its stunning graphics capabilities. Both of these fascinating graphics packages work perfectly on the Electron and will help you to make the most of these capabilities.



TRACKMAN

This game of thought and co-ordination is an excellent one for ELBUG members. Although it will run as it stands on the Electron it is rather slow. By converting the whole game to run in mode 5 we can speed things up considerably without any real losses. Here are the changes suggested:

```
140 REPEAT MODE1
170 PROCcb
180 PROCT:MODE 5
1440 COLOURT%:IF T%=2 COLOUR1
2140 VDU19,0,4,0,0,0,19,1,3,0,0,0
2670 PRINTTAB(0,25);:VDU17,0,17,129
```

Finally, delete line 2150

HINTS

The following hints in this month's BEEBUG are applicable to ELBUG members.

THE VALUE OF THE £
SINGLE KEY STRING SEARCH
VDU21 ANOMALY
INPUT ERRORS

Cumana Disc Interface

Hot on the heels of the official Acorn Plus 3 disc drive and interface, comes Cumana's disc interface for the Electron. Alan Webster investigates.

The Cumana disc interface will cost you £115 but also requires you to own an Acorn Plus 1 unit. Plugging into the Plus 1 expansion, the interface allows the connection of any one, or even two, of the many standard 5.25, 3.5 or 3 inch disc drives available for the BBC micro.


The interface is a small, slim box measuring 146 mm by 113 mm by 22 mm which plugs into either of the cartridge sockets on the Plus 1, but preferably into the back one so that the other is available for other software. I found that because the unit projects vertically out of the Plus 1, it tends to come loose if knocked and it also got in the way of the monitor unless the monitor was on a stand. However, the same applies to any Plus 1 cartridge.

The interface contains its own disc filing system (DFS) on a 16K EPROM and a battery-backed real time clock. The DFS contains most of the software that you need when operating discs and using files.

The real time clock is a useful addition that will, for once, put your Electron above a BBC micro. It keeps a record of the time and date and records these with each disc file so that you know when a program was last changed. The clock is set with a program included on the utilities disc but there is only need to do this infrequently as the system has its own built-in battery to keep the clock ticking over even when the unit is not plugged in, or the Electron is switched off.

The clock setting program can also be used to give a running display of the time. Accessing the time and date from within your own programs is more difficult, you have to resort to machine code. It is a pity that Cumana couldn't have included a command along the lines of *TIME that would print out the time and date at the present text co-ordinates.

When you receive your disc interface you will receive a 70 page A5 guide which



will show you how to use the disc commands and explain a little about data storage and what discs are. You will also receive a disc containing various utilities.

The Cumana disc interface stores information on discs in a different format to that used on both the BBC micro and the Electron with Acorn's Plus 3. However, the system is closer in many ways to that available on the BBC micro than the Plus 3.

The utilities disc contains a disc formatter to prepare discs for use in both Cumana and Acorn disc systems and utilities for copying one or more files from Acorn BBC or Electron formatted discs to Cumana ones, and vice versa. However, you will require a 3.5 in disc drive to read Electron Plus 3 discs and a 5.25 in drive to read BBC micro discs. There are also utilities on the disc for copying from double to single density, for editing a disc and to verify a disc.

The main commands, built into the DFS, that will be new to you are:

| | |
|---------|--|
| *ACCESS | This 'Locks' or 'Unlocks' a file. If a file is 'Locked' then it cannot be accidentally overwritten or erased |
|---------|--|

***BOOT** This allows commands to be executed when Break is pressed. It is similar in operation to defining the Break key.

***BUILD** This command sets up a file that can be EXEC'ed back into memory. This is most useful for creating a file on disc called !BOOT that will be executed when Shift Break is pressed.

***CAT** will produce a catalogue of the disc on screen, giving all the file names and their relevant information.

***CLOSE** will close all of the currently opened files on disc. It is similar to CLOSE#0 in Basic but this command can be used when you are not in Basic.

***DELETE** This command can be used to delete one, or a number of files from the disc.

***DIR** This command will select the current drive and directory, or set the directory only. For example, if you wish to load a file called Z.ELK from drive 2 then the commands *DIR:2.Z followed by LOAD "ELK" will work. This is useful if you wish to use a lot of files in one directory.

***DRIVE** Sets the current drive number. If you are using the reverse side of a disc a lot, then *DRIVE 1 will be useful.

***DUMP** Produces an ASCII listing of the chosen disc file on screen.

***FREE** This gives information on the number of files and the amount of free sectors on the disc.

***INFO** This produces the catalogue information on certain files, giving the load address, execution address, length and the date when the file was last updated.

***LIB** This command in most respects is similar to *DIR except that *LIB is used to redirect the machine operating system calls.

***OPT** This selects the options that you require when loading a file and sets the auto-boot option.

***PBOOT** Prints out the option set by *OPT

***RENAME** This command changes the name of a file to another name.

***TITLE** This allows you to set the title for a given disc. The title is always displayed when a *CAT command is issued.

***TYPE** This command will list a file that has been stored using *SPOOL or *BUILD.

There is one small drawback when using disc drives on the Electron, and that is that some of the cheaper disc drives do not have their own power supply. The BBC micro has a connection so that disc drives can use its power supply, but no such connections exist on the Electron. So when you buy a disc drive, make sure it has its own power supply.

Overall, I found the disc interface a vast improvement over using cassette. The speed improvement and general handiness of having easy access to each file, and knowing where a certain program is a godsend for the Electron. If you already have a Plus 1 interface, or you are even considering getting one, the Cumana disc interface is a really worthwhile investment.